Project Work

E-commerce(Electronics and Gadgets) site ontology

Domain:

E-commerce website, selling technological products (cell phones, pc, accessories, videogames,...)

Subdomains:

- Product Domain (hierarchical organization in categories, properties,...)
- Customer activity Domain (clicks on products, buying, reviews,...)

Goals:

- Improve products retrieval and customer experience
- Customer segmentation (infer new customer classes based on their activities)
- Query for similar customers and similar products (co-view, co-buy)

Project Work

E-commerce(Electronics and Gadgets) site ontology

Key Design Choices in Building an E-Commerce Ontology

1. Domain Scope
Description: This ontology spans diverse e-commerce categories.
Example Aspects:

- Products: Electronics, Gaming, Informatics
- User Interactions: Buying, Reviewing, Visualizing

2. Class vs. Individual
Classes: Broad categories (e.g., 'Product', 'UserAction', 'Location').
Individuals: Specific entities (e.g., 'NintendoSwitch', 'ProductReview by Alice', 'Paris').
Application: Enables detailed representation of products and user activities.

3. Data vs. Object Properties
Data Properties: Specific attributes (e.g., price of 'PlayStation4', rating in 'ProductReview').
Object Properties: Relationships (e.g., 'ProductBuying' linked to both 'Product' and 'Person').
Usage: Facilitates rich, interconnected data representation.

Project Work

E-commerce(Electronics and Gadgets) site ontology

Key Design Choices in Building an E-Commerce Ontology

4. N-ary Relationships
Description: Captures complex multi-entity interactions.
Example: 'ProductBuying' involving a 'Person', a 'Product', and a 'Location'.
Implementation: Enhances the depth of user interaction modeling.

5. Using Existing Vocabularies for Interoperability
Vocabularies: Incorporates standards like **FOAF, VCARD, GoodRelations**.
Purpose: Ensures broad compatibility and ease of data integration.
Benefit: Makes the ontology versatile and widely applicable.

Tool: protégé

- Popular
- Many features and plugins(different visualizations, many query languages supported, many reasoners available.)
- Easy to use (graphic interface, many views)

Project Work

E-commerce(Electronics and Gadgets) site ontology

Querying and Reasoning

- **DL(Description Logic)** Queries
- Used **Pellet** reasoner (**support for Owl DL and SWRL**)

LOW COST GAMING PRODUCT

Product_Gaming **and** hasPrice **some** xsd:double[< 25.0]

CUSTOMERS WHO LIKED PRODUCT X

performsAction **some** (ProductReview **and** reviewRating **value** 5 **and** ofProduct **value** B08BPTKHJH)

GAMER CUSTOMER

→ **segmentation by Product Category**

performsAction **some** (ofProduct **some** Videogame) **and** (performsAction **some** (ofProduct **some** VideogameConsole) **or** performsAction **some** (ofProduct **some** VideogameAccessory))

Project Work

E-commerce(Electronics and Gadgets) site ontology

Querying and Reasoning

- ❑ **SPARQL** Queries
- ❑ Plugins: SPARQL query

We will always use these prefixes:

PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX gr: <http://purl.org/goodrelations/v1#>
PREFIX vcard: <http://www.w3.org/2006/vcard/ns#>
PREFIX : <http://www.semanticweb.org/ontologies/2022/e-commerce#>

Project Work

E-commerce(Electronics and Gadgets) site ontology

Querying and Reasoning

- **SPARQL** Queries
- Plugins: SPARQL query

GET ALL PRODUCTS

```
SELECT DISTINCT ?p ?category
WHERE {
?p a ?category.
?category(rdfs:subClassOf)+:Product.
}
```

USER(Cristian_the_gamer) HISTORY

```
SELECT ?datetime ?action ?product
WHERE {
?a :performedByUser :Cristian_the_gamer;
            a ?action;
           :actionDatetime ?datetime;
           :ofProduct ?p.
?p gr:name ?product.
FILTER (?action != owl:NamedIndividual)
 }
ORDER BY ?datetime
```

Project Work

E-commerce(Electronics and Gadgets) site ontology

Querying and Reasoning

- ❑ **SPARQL** Queries
- ❑ Plugins: SPARQL query

PRODUCT'S INFORMATION RETRIEVAL

```
SELECT DISTINCT ?property ?value
WHERE {
:B079Z3V9WT ?property ?value
FILTER ( ?property != rdf:type )
}
```

RELATED PRODUCTS OF B079Z3V9WT

→ improved product retrieval

→ recommendations

```
SELECT DISTINCT ?productID ?name ?price ?brand ?category
WHERE { #Products of the same category
{:B079Z3V9WT a ?category.
?productID a ?category;
        gr:name ?name;
        gr:hasBrand ?brand;
        :hasPrice ?price.
FILTER (?productID != :B079Z3V9WT ) }
UNION #Products of the same brand
{?productID a ?category;
         gr:name ?name;
         gr:hasBrand ?brand;
         :hasPrice ?price.
:B079Z3V9WT gr:hasBrand ?brand. }
FILTER (?category != owl:NamedIndividual) }
```

Project Work

E-commerce(Electronics and Gadgets) site ontology

Querying and Reasoning

- **SPARQL** Queries
- Plugins: SPARQL query

VERIFIED PURCHASE

```
SELECT ?review ?user ?product
WHERE {
?review a :ProductReview;
        :ofProduct ?product;
        :performedByUser ?user.
?buy a :ProductBuying;
        :ofProduct ?product;
        :performedByUser ?user.
}
```

BEST SELLERS

```
SELECT DISTINCT ?product_name
        (COUNT (?buy) AS ?n_buying)
WHERE {
 ?buy a :ProductBuying;
        :ofProduct ?p.
?p gr:name ?product_name
}
GROUP BY ?product_name
HAVING (?n_buying > 1)
ORDER BY DESC (?n_buying)
```

Project Work

E-commerce(Electronics and Gadgets) site ontology

Querying and Reasoning

- **SPARQL** Queries
- Plugins: SPARQL query

POPULAR BRANDS

```
SELECT DISTINCT ?brand
        (COUNT (?buy) AS ?n_buying)
WHERE {
?p gr:hasBrand ?brand.
?buy a :ProductBuying;
        :ofProduct ?p
}
GROUP BY ?brand
HAVING (?n_buying > 1)
ORDER BY DESC (?n_buying)
```

MOST BOUGHT CATEGORIES

```
SELECT DISTINCT ?category
        (COUNT (?buy) AS ?n_buying)
WHERE {
?p a ?category.
FILTER (?category != owl:NamedIndividual)
?buy a :ProductBuying;
        :ofProduct ?p
}
GROUP BY ?category
HAVING (?n_buying > 1)
ORDER BY DESC (?n_buying)
```